
Matrix LET Subcommands MATRIX TRANSPOSE

DATAPLOT Reference Manual March 19, 1997 4-55

MATRIX TRANSPOSE

PURPOSE
Compute the transpose of a matrix.

DESCRIPTION
The transpose of a matrix is another matrix in which the rows and columns have been reversed. For example,

(EQ 4-71)

would have the transpose--

(EQ 4-72)

SYNTAX
LET <mat2> = MATRIX TRANSPOSE <mat1> <SUBSET/EXCEPT/FOR qualification>
where <mat1> is a matrix for which the transpose is to be computed;

 <mat2> is a matrix where the resulting transpose is saved;
and where the <SUBSET/EXCEPT/FOR qualification> is optional and rarely used in this context.

EXAMPLES
LET C = MATRIX TRANSPOSE A

NOTE 1
The row and column dimensions for the transposed matrix are the opposite dimensions of the original matrix. This means that the
number of rows in the original matrix cannot be larger than the column limit for matrices (100 in the current implementation).

NOTE 2
DATAPLOT can compute column statistics for matrices fairly easily. For example, to compute the means of the 5 columns of matrix M,
do the following:

LOOP FOR K = 1 1 5
LET MEAN^K = MEAN M^K

END OF LOOP

However, DATAPLOT does not compute statistics for rows directly. The MATRIX TRANSPOSE command can be used for this
purpose. The columns of the transpose correspond to the rows of the original matrix. For example, to compute the row means for the
above matrix M, do the following:

LET MT = MATRIX TRANSPOSE M
LET N = SIZE MT1
LOOP FOR K = 1 1 N

LET ROWM^K = MEAN MT^K
END OF LOOP

The program example shows how this can be used to compute a chi-square. The CROSS TABULATE command does it directly.

DEFAULT
None

SYNONYMS
None

RELATED COMMANDS
MATRIX DEFINITION = Set a matrix definition.
MATRIX INVERSE = Compute a matrix inverse.

MATRIX NUMBER OF COLUMNS = Compute the number of columns in a matrix.
MATRIX NUMBER OF ROWS = Compute the number of rows in a matrix.
MATRIX RANK = Compute the rank of a matrix.
MATRIX TRACE = Compute a matrix trace.

REFERENCE
Any standard text on linear algebra.

APPLICATIONS
Linear Algebra

IMPLEMENTATION DATE
87/10

PROGRAM
. COMPUTE CHI-SQUARE TEST FOR FOLLOWING FREQUENCY MATRIX:
READ MATRIX M
14 37 32
19 42 17
12 17 10
END OF DATA
LET NROW = SIZE M1
LET NCOL = MATRIX NUMBER OF COLUMNS M
.
LOOP FOR K = 1 1 NCOL

LET COLT^K = SUM M^K
END OF LOOP
.
LET MT = MATRIX TRANSPOSE M
LOOP FOR K = 1 1 NROW

LET TEMP = SUM MT^K
LET ROWT(K) = TEMP

END OF LOOP
LET GT = SUM ROWT
.
LET CHISQ = 0
LOOP FOR K = 1 1 NCOL

LET E = ROWT*(COLT^K/GT)
LET E = (M^K - E)**2/E
LET A = SUM E
LET CHISQ = CHISQ + A

END OF LOOP
.
LET DF = (NROW-1)*(NCOL-1)
LET CV = CHSPPF(.95,DF)
PRINT “THE CHI-SQUARE TEST STATISTIC = ^CHISQ”
PRINT “THE CRITICAL VALUE (ALPHA = .05) = ^CV”

The following output is generated.

THE CHI-SQUARE TEST STATISTIC = 7.464394
THE CRITICAL VALUE (ALPHA = .05) = 9.487729


